
・ PLCの内臓Ethernetポート設定

・ PLCの確認はデバイス/バッファメモリ一括モニタで行いました。

MPC-3000/3200 Q3Eコマンド サンプル
ti2k-220202(Last Update 220601)

MPC-3000(SH2A) BL/I 2.01_50 2021/09/03

MELSOFTシリーズ GX Works2

■ TCP/IP通信

Q03UDECPU
(スレーブ)

192.168.0.195
TCP MCプロトコル
ポート番号 9000

MPC-3000/3200

SET_IP TCP Telnet
SET_IP 192 168 0 20 255 255 255 0 192 168 0 248
SET_IP TCP PACKET 9000 192 168 0 195   /* Q03UDECPUがスレーブ

ON_ERROR  *sys_err

/* どれか一つアンコメントアウトして実行
QUIT_FORK 1 *WR_BULK    /* WRITE_BULK, READ_BULK example
/*QUIT_FORK 1 *WR_RNDM    /* WRITE_RNDM, READ_RNDM example
/*QUIT_FORK 1 *W_BIT       /* WRITE_BIT example

END

*WR_BULK
d=0
w=0
m=0
x=0
y=0
b=0
DO
  d=d+1
  X(100)=d
  X(101)=d+1
  'X(100)下位2バイト -> Q03 D100 (Q03のアドレスはワードのDEC表記)
  'X(101)下位2バイト -> Q03 D101
  Q3E TCP|0 WRITE_BULK "D" 100 X(100) 2
  'Y(100)下位2バイト <- Q03 D100
  'Y(101)下位2バイト <- Q03 D101
  Q3E TCP|0 READ_BULK  "D" 100 Y(100) 2
  PR "DW" X(100) X(101)
  PR "DR" Y(100) Y(101)

  w=w+1
  X(200)=w
  X(201)=w+1
  'X(200)下位2バイト -> Q03 W0C8 (Q03のアドレスはワードのHEX表記)
  'X(201)下位2バイト -> Q03 W0C9
  Q3E TCP|0 WRITE_BULK "W" 200 X(200) 2
  'Y(200)下位2バイト <- Q03 W0C8
  'Y(201)下位2バイト <- Q03 W0C9
  Q3E TCP|0 READ_BULK  "W" 200 Y(200) 2
  PR "WW" X(200) X(201)
  PR "WR" Y(200) Y(201)

  m=m+1
  X(300)=m
  X(301)=m+1
  'X(300)下位2バイト -> Q03 M300 (Q03のアドレスはビットのDEC表記)
  'X(301)下位2バイト -> Q03 M316
  Q3E TCP|0 WRITE_BULK "M" 300 X(300) 2
  'Y(300)下位2バイト <- Q03 M300
  'Y(301)下位2バイト <- Q03 M316
  Q3E TCP|0 READ_BULK  "M" 300 Y(300) 2
  PR "MW" X(300) X(301)
  PR "MR" Y(300) Y(301)

  x=x+1
  X(400)=x
  X(401)=x+1
  'X(400)下位2バイト -> Q03 X190 (Q03のアドレスはビットのHEX表記)
  'X(401)下位2バイト -> Q03 X1A0
  Q3E TCP|0 WRITE_BULK "X" 400 X(400) 2
  'Y(400)下位2バイト <- Q03 X190
  'Y(401)下位2バイト <- Q03 X1A0
  Q3E TCP|0 READ_BULK  "X" 400 Y(400) 2
  PR "XW" X(400) X(401)
  PR "XR" Y(400) Y(401)

  y=y+1
  X(500)=y
  X(501)=y+1
  'X(500)下位2バイト -> Q03 Y1F4 (Q03のアドレスはビットのHEX表記)
  'X(501)下位2バイト -> Q03 Y204
  Q3E TCP|0 WRITE_BULK "Y" 500 X(500) 2
  'Y(500)下位2バイト <- Q03 Y1F4
  'Y(501)下位2バイト <- Q03 Y204
  Q3E TCP|0 READ_BULK  "Y" 500 Y(500) 2
  PR "YW" X(500) X(501)
  PR "YR" Y(500) Y(501)

  b=b+1
  X(600)=b
  X(601)=b+1
  'X(600)下位2バイト -> Q03 B258 (Q03のアドレスはビットのHEX表記)
  'X(601)下位2バイト -> Q03 B268
  Q3E TCP|0 WRITE_BULK "B" 600 X(600) 2
  'Y(600)下位2バイト <- Q03 B258
  'Y(601)下位2バイト <- Q03 B268
  Q3E TCP|0 READ_BULK  "B" 600 Y(600) 2
  PR "BW" X(600) X(601)
  PR "BR" Y(600) Y(601)

  TIME 1000
LOOP

*WR_RNDM

d=0
DO
  d=d+1
  'Q3E の W3E8,W3F0,W3F8 に書き込み
  Q3E TCP|0 WRITE_RNDM "W" 1000 d "W" 1008 d+1 "W" 1016 d+2
  /*Q3E TCP|0 WRITE_RNDM "W" &H3E8 d "W" &H3F0 d+1 "W" &H3F8 d+2 /* 前文と同じ
  'Q3E の W3E8,W3F0,W3F8 から MBK(100)~(102)に読み込み
  Q3E TCP|0 READ_RNDM  "W" 1000 "W" 1008 "W" 1016 MBK(100)
  PR MBK(100) MBK(101) MBK(102)
  TIME 1000
LOOP

*W_BIT
  c=0
  DO
    c=c+1
    c=c%2
    Q3E TCP|0 WRITE_BIT "X" &H100 1 c         /* X100 on/off
    Q3E TCP|0 WRITE_BIT "Y" &H100 2 c 1-c     /* Y100,101 on/off
    Q3E TCP|0 WRITE_BIT "B" &H10C 3 1-c c 1-c /* B10C,10D,10E on/off
    Q3E TCP|0 READ_BULK "X" &H100 MBK(200) 1
    Q3E TCP|0 READ_BULK "Y" &H100 MBK(201) 1
    Q3E TCP|0 READ_BULK "B" &H100 MBK(202) 1
    PRX MBK(200) MBK(201) MBK(202)
    TIME 500
  LOOP

*sys_err
  PRINT "TCP(0)=" TCP(0)
  PRINT "ERROR Line:" err_&&H00FFFFFF "Code:" err_>>24 "Message:" ERR$(err_)
  END

・ MPCプログラム

RUN

# DW 1 2
 DR 1 2
 WW 1 2
 WR 1 2
 MW 1 2
 MR 1 2
 XW 1 2
 XR 1 2
 YW 1 2
 YR 1 2
 BW 1 2
 BR 1 2
 DW 2 3
 DR 2 3
 WW 2 3
 WR 2 3
 MW 2 3
 MR 2 3
 

・ プログラム実行結果

RUN

# 1 2 3
 2 3 4
 3 4 5
 4 5 6
 5 6 7
 6 7 8
 7 8 9
 8 9 10
 9 10 11
 10 11 12
 11 12 13
 12 13 14
 13 14 15
 14 15 16
 15 16 17
 16 17 18
 17 18 19
 18 19 20

RUN

#00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 
00000001 00000001 00002000 
00000000 00000002 00005000 

*WR_BULKタスク     *WR_RNDMタスク     *W_BITタスク 



Dデバイスはワード(2バイト)の10進表記

Wデバイスはワード(2バイト)の16進表記

X,Y,Bデバイスはビットの16進表記

Mデバイスはビットの10進表記

■ PLCのデバイスアドレス指定の注意点
デバイスによりワード/ビット、10進/16進が異なります。

■ UDP/IP通信

SET_IP TCP Telnet
SET_IP 192 168 0 20 255 255 255 0 192 168 0 248

/* UDP設定
SET_IP UDP PACKET 10000       /* Q3Eのポート番号
IP_CONV 192 168 0 195 DestIP  /* Q3EのIPアドレス→変数DestIP
SET_DEST DestIP UDP_X0$       /* UDP Ch0の宛先を設定

ON_ERROR  *sys_err

・ MPC UDP通信の設定 (2ページのプログラムの冒頭部分)

書き込み
Q3E UDP|0 WRITE_BULK "D" 100 X(100) 2
Q3E UDP|0 WRITE_RNDM "W" 1000 d "W" 1008 d+1 "W" 1016 d+2
Q3E UDP|0 WRITE_BIT "X" &H100 1 c 

読み込み
Q3E UDP|0 READ_BULK  "D" 100 Y(100) 2
Q3E UDP|0 READ_RNDM  "W" 1000 "W" 1008 "W" 1016 MBK(100)

・コマンド

・ PLCの内臓Ethernetポート設定

Q03UDECPU
(スレーブ)

192.168.0.195
UDP MCプロトコル
ポート番号 10000



■TCP タイムアウトオプション(MPC-3000(SH2A) BL/I 2.01_48 2021/02/17 以降)。

オプション無し タイムアウト 10 秒 
Q3E TCP|p WRITE_BULK "D" 1000 aho(50) 3

タイムアウト n=1～30 [秒] 
Q3E (n,TCP|p) WRITE_BULK "D" 1000 aho(50) 3

エラーコード
88 「TCP/IPが切断されています」
90 「TCP_XMTタイムアウトエラー」
91 「TCP_RCVタイムアウトエラー」

・MPCプログラム例

SET_IP TCP Telnet
SET_IP 192 168 0 20 255 255 255 0 192 168 0 248
SET_IP TCP PACKET 9000 192 168 0 195

ON_ERROR  *on_err

WAIT TCP(0)==1
d=0
DO
  d=d+1
  FILL X(400) 100 d 1
  Q3E TCP|0 WRITE_BULK "D" 400 X(400) 100   /* X(400)~100 word -> Q03 D400~
  TIME 50
  Q3E TCP|0 READ_BULK  "D" 400 Y(400) 100    /* Y(400)~ <- Q03 D400~100 word
  TIME 50
LOOP

*on_err
 errmsg$=TIME$(1)+" Line:"+STR$(err_&&H00FFFFFF)+" Code:"+STR$(err_>>24)+" "+ERR$(err_)
 PR errmsg$
 RESUME
 END
 
・実行結果
 
RUN   (実行中にMPCのEthernetケーブルを抜いてみる) 

# 11:22:24 Line:110 Code:90 TCP_XMTタイムアウトエラー
 11:22:39 Line:110 Code:91 TCP_RCVタイムアウトエラー
 11:22:49 Line:110 Code:90 TCP_XMTタイムアウトエラー

(Ethernetケーブルを差しても通信は回復しない。PLCのEthernet診断の「TCP状態」は「切断」。MPCパワーオンリセットで回復。)

■UDP タイムアウトオプション(MPC-3000(SH2A) BL/I 2.01_48 2021/02/17 以降)。

オプション無し タイムアウト 10 秒 
Q3E UDP|p WRITE_BULK "D" 1000 aho(50) 3

タイムアウト n=1～30 [秒] 
Q3E (n,UDP|p) WRITE_BULK "D" 1000 aho(50) 3

エラーコード (UDPでも TCP と表示されます)
88 「TCP/IPが切断されています」
90 「TCP_XMTタイムアウトエラー」
91 「TCP_RCVタイムアウトエラー」

・プログラム例

SET_IP TCP Telnet
SET_IP 192 168 0 20 255 255 255 0 192 168 0 248

/* UDP設定
SET_IP UDP PACKET 10000       /* Q3Eのポート番号
IP_CONV 192 168 0 195 DestIP  /* Q3EのIPアドレス→変数DestIP
SET_DEST DestIP UDP_X0$       /* UDP Ch0の宛先を設定

ON_ERROR  *on_err

d=0
DO
  d=d+1
  FILL X(400) 100 d 1
  Q3E (3,UDP|0) WRITE_BULK "D" 400 X(400) 100   /* X(400)~100 word -> Q03 D400~
  TIME 50
  Q3E (3,UDP|0) READ_BULK  "D" 400 Y(400) 100    /* Y(400)~ <- Q03 D400~100 word
  TIME 50
LOOP

*on_err
 errmsg$=TIME$(1)+" Line:"+STR$(err_&&H00FFFFFF)+" Code:"+STR$(err_>>24)+" "+ERR$(err_)
 PR errmsg$
 RESUME
 END

・実行結果

RUN (実行中にMPCのEthernetケーブルを抜いてみる)

# 14:03:35 Line:120 Code:91 TCP_RCVタイムアウトエラー
 14:03:38 Line:120 Code:91 TCP_RCVタイムアウトエラー
 14:03:41 Line:120 Code:91 TCP_RCVタイムアウトエラー
 14:03:44 Line:120 Code:91 TCP_RCVタイムアウトエラー

(Ethernetケーブルを差すと通信が回復した)

・MPCプログラムの設定部分

SET_IP TCP Telnet
SET_IP 192 168 0 20 255 255 255 0 192 168 0 248
/* UDP設定
SET_IP UDP PACKET 10000       /* Q3Eのポート番号
IP_CONV 192 168 0 195 DestIP  /* Q3EのIPアドレス→変数DestIP
SET_DEST DestIP UDP_X0$       /* UDP Ch0の宛先を設定

・SET_IPコマンドで確認

#SET_IP
IPadrs = 192.168.0.20
 MASK  = 255.255.255.0
Gateway= 192.168.0.248
OFF Delay ACK  baud=115200
MACadrs= 001EC08DAB6E

1 : /  TCP   Telnet          23  Slave 
2 : +  UDP   PACKET       10000 <UDP_X0$><UDP_R0$>
#

PLCポート番号 この数字がコマンドと整合しているか

Q3E (3,UDP|0) WRITE_BULK "D" 400 X(400) 100

■Ethernet設定/通信状況の確認。MPC、PLCとも正常に通信が出来ている状態です。

・PLCの状況 (変更したらRESETか電源入れ直しが必要?)

※ MPCは設定プログラムを変更したら初期化してパワーオンリセット
例) 
#RUN IP_INIT ←RUN IP_INIT 実行
Eraes IP settings
Refresh IP settings

All the tasks have been quit !!
Wake TCP 
 ID:01 :  TASK 46 :  Telnet 

MKY =0                                         ←電源入れなおす
+++++++++++++++++
MMC_card Mounted
Detect = 24AA02E48 
IPadrs = 192.168.0.20
 MASK  = 255.255.255.0
Gateway= 192.168.0.248
OFF Delay ACK  baud=115200
MACadrs= 001EC08DAB6E
Ether Opened
TCP_STACK_RAM=11768
TCP Init Completed
Wake TCP 
 ID:01 :  TASK 46 :  Telnet 
Wake UDP 
 ID:02 :   TASK 45 :  UDP PACKET <UDP_X0$><UDP_R0$>
Wake Serial
USB_I/F  Ver1.13
iMPC-3000 
#



■Q3E UDP コマンド仕様追加

MPC-3x00(SH2A) BL/I 2.01_52 2022/03/29

1) エラーリターン変数追加

UDPではエラーが多くなるので、エラーリターン機能を追加しました。
引数の末尾に変数を追加するとその変数にエラーコードが入ります。
変数がなければ従来通り ランタイムエラーで停止します(RESUME復旧)。

  エラーコード 94:送出エラー, 95:受信エラー

2)レスポンスタイム参照

PEEK(UDP_R0$+501)でPLDからの応答時間が得られます。

・プログラム例

SET_IP 192 168 0 20 255 255 255 0 192 168 0 248
SET_IP TCP Telnet

/* UDP設定
SET_IP UDP PACKET 10000       /* Q3Eのポート番号
IP_CONV 192 168 0 195 DestIP  /* Q3EのIPアドレス→変数DestIP
SET_DEST DestIP UDP_X0$       /* UDP Ch0の宛先を設定

d=0
DO
  d=d+1
  FILL X(400) 100 d 1
  DO
    starttime=SYSCLK                                    /* コマンド実行の所要時間の測定
    Q3E (1,UDP|0) WRITE_BULK "D" 400 X(400) 100 Q3Err   /* X(400)~100 word -> Q03 D400~
    elapsedtime=SYSCLK-starttime
    IF Q3Err==0 THEN                                    /* エラーチェック
      BREAK
    END_IF
    PR "WRITE_BULK Err" Q3Err
  LOOP
  PR "WRITE_BULK Ok" PEEK(UDP_R0$+501) elapsedtime
  TIME 100
  DO
    starttime=SYSCLK
    Q3E (1,UDP|0) READ_BULK  "D" 400 Y(400) 100 Q3Err   /* Y(400)~ <- Q03 D400~100 word
    elapsedtime=SYSCLK-starttime
    IF Q3Err==0 THEN
      BREAK
    END_IF
    PR "READ_BULK Err" Q3Err
  LOOP
  PR "READ_BULK Ok" PEEK(UDP_R0$+501) elapsedtime
  TIME 100
LOOP

・実行結果

WRITE_BULK Ok 7 10
READ_BULK Ok 7 10
WRITE_BULK Ok 8 11
READ_BULK Ok 7 9
WRITE_BULK Ok 7 10  ←PLC のEthernetケーブルを抜いてみる
READ_BULK Err 95
READ_BULK Err 95
READ_BULK Err 95    ←ケーブルを差す
READ_BULK Ok -14 800
WRITE_BULK Ok 7 57
READ_BULK Ok 6 10
WRITE_BULK Ok 6 10
READ_BULK Ok 6 10
WRITE_BULK Ok 7 10
READ_BULK Ok 3 10
WRITE_BULK Ok 6 60
READ_BULK Ok 7 55
WRITE_BULK Ok 7 10
READ_BULK Ok 7 10
WRITE_BULK Ok 7 10
READ_BULK Ok 7 10   ←PLC のEthernetケーブルを抜いてみる
WRITE_BULK Err 95
WRITE_BULK Err 95
WRITE_BULK Err 95   ←ケーブルを差す
WRITE_BULK Ok 87 390
READ_BULK Ok 6 17
WRITE_BULK Ok 7 55
READ_BULK Ok 7 10
WRITE_BULK Ok 7 10

UDPの状態は経路が成立していても非通信状態では 「+」 になります。

SET_IP
IPadrs = 192.168.0.20
 MASK  = 255.255.255.0
Gateway= 192.168.0.248
OFF Delay ACK  baud=38400
MACadrs= 70B3D553D051

1 : =  TCP   Telnet          23  Slave  connected-> 192.168.0.78
2 : +  UDP   PACKET       10000 <UDP_X0$><UDP_R0$>

コマンド実行中だけ「=」 になります(極短時間)。

SET_IP
IPadrs = 192.168.0.20
 MASK  = 255.255.255.0
Gateway= 192.168.0.248
OFF Delay ACK  baud=38400
MACadrs= 70B3D553D051

1 : =  TCP   Telnet          23  Slave  connected-> 192.168.0.78
2 : =  UDP   PACKET       10000 <UDP_X0$><UDP_R0$>

■UDPのSET_IP状態  (BL/I 2.01_54 2022/04/07)

・Q3Eコマンド動作確認

Q3E UDP|0 WRITE_RNDM "W" 1000 987 "W" 1008 654 "W" 1016 321 L050ErrRet  /* Ok
Q3E UDP|0 WRITE_RNDM "W" 1000 111 "W" 1008 222 "W" 1016 333             /* Ok

Q3E (1,UDP|0) WRITE_RNDM "W" 1000 1234 "W" 1008 5678 "W" 1016 9876 L050ErrRet /* Ok
Q3E (1,UDP|0) WRITE_RNDM "W" 1000 4321 "W" 1008 8765 "W" 1016 6789            /* Ok

Q3E (1,UDP|0) WRITE_RNDM "B" &H1FF0 1234                                /* Ok
Q3E (1,UDP|0) WRITE_RNDM "B" &H1FF0 4321 L050ErrRet                     /* Ok
Q3E (1,UDP|0) WRITE_RNDM "B" 8222 300                                   /* 8222は範囲外 エラー変数無し
  MCプロトコルエラーリターン: 82                                        /* ランタイム エラーメッセージ

L050ErrRet=0
Q3E (1,UDP|0) WRITE_RNDM "B" 8222 300 L050ErrRet                        /* 範囲外 エラー変数付加
PR L050ErrRet                                                           /* エラー変数にエラーコードが入る
 82

Q3E (1,UDP|0) READ_RNDM "W" 1000 "W" 1008 "W" 1016 MBK(100)             /* Ok
Q3E (1,UDP|0) READ_RNDM "W" 1000 "W" 1008 "W" 1016 MBK(100) L050ErrRet  /* Ok

Q3E (1,UDP|0) READ_RNDM "B" &H1FF0 MBK(200)                             /* Ok
Q3E (1,UDP|0) READ_RNDM "B" &H1FF0 MBK(200) L050ErrRet                  /* Ok

Q3E (1,UDP|0) READ_RNDM "B" 8222 MBK(200)                               /* 8222は範囲外 エラー変数無し
  MCプロトコルエラーリターン: 82                                        /* ランタイム エラーメッセージ

L050ErrRet=0
Q3E (1,UDP|0) READ_RNDM "B" 8222 MBK(200) L050ErrRet                    /* 範囲外 エラー変数付加
PR L050ErrRet                                                           /* エラー変数にエラーコードが入る 
 82    

reg$="W"                                                                /* レジスタの文字列指定は不可
Q3E (1,UDP|0) WRITE_RNDM reg$ 1000 1234 reg$ 1008 5678 reg$ 1016 9876 L050ErrRet
None device pointer !!
Q3E (1,UDP|0) WRITE_RNDM reg$ 1000 1234 reg$ 1008 5678 reg$ 1016 9876
None device pointer !!

Q3E (1,UDP|0) READ_RNDM reg$ 1000 reg$ 1008 reg$ 1016 MBK(100) L050ErrRet
None device pointer !!
Q3E (1,UDP|0) READ_RNDM reg$ 1000 reg$ 1008 reg$ 1016 MBK(100)
None device pointer !!
                                                                 



■PCでエラーを記録する

7ページのプログラムにエラーをLogするPCを追加しました。
Q3Eエラー発生時にコマンドとコード、毎正10分に通信回数をLogします。

・プログラム

SET_IP TCP Telnet
SET_IP 192 168 0 20 255 255 255 0 192 168 0 248

/* Q3E(PLD) UDP
SET_IP UDP PACKET 10000       /* ポート番号
IP_CONV 192 168 0 195 q3eIp   /* IPアドレス→変数q3eIp
SET_DEST q3eIp UDP_X0$        /* Ch0の宛先を設定

/* PC UDP  add 220527
SET_IP UDP PACKET 9000        /* ポート番号
IP_CONV 192 168 0 71 pcIp     /* IPアドレス→変数pcIp
SET_DEST pcIp UDP_X1$         /* Ch1の宛先を設定

timeFlg=0
jobCount=0
errCount=0
sendStr$=TIME$(1)+" Begin"
GOSUB *sendLog
DO
  jobCount=jobCount+1
  FILL X(400) 100 jobCount 1
  DO
    startTime=SYSCLK                                      /* コマンド実行の所要時間の測定
    Q3E (1,UDP|0) WRITE_BULK "D" 400 X(400) 100 errorCode /* X(400)~100 word -> Q03 D400~
    elapsedTime=SYSCLK-startTime
    IF errorCode==0 THEN                                  /* エラーチェック
      BREAK
    END_IF
    process$="WRITE_BULK"
    GOSUB *errorInfo
  LOOP
  PR "WRITE_BULK Ok" PEEK(UDP_R0$+501) elapsedTime
  TIME 100
  DO
    startTime=SYSCLK
    Q3E (1,UDP|0) READ_BULK  "D" 400 Y(400) 100 errorCode  /* Y(400)~ <- Q03 D400~100 word
    elapsedTime=SYSCLK-startTime
    IF errorCode==0 THEN
      BREAK
    END_IF
    process$="READ_BULK"
    GOSUB *errorInfo
  LOOP
  PR "READ_BULK Ok" PEEK(UDP_R0$+501) elapsedTime
  TIME 100
  IF (TIME(0)&&H00000F00==0)&(timeFlg==0) THEN            /* 毎正10分
    sendStr$=TIME$(1)+" "+STR$(jobCount)+"回"             /* 時刻とQ3E通信回数
    GOSUB *sendLog
    timeFlg=1
  END_IF
  IF TIME(0)&&H00000F00<>0 THEN
    timeFlg=0
  END_IF
LOOP

*errorInfo
  PR process$ "Error" errorCode
  errCount=errCount+1
  sendStr$=TIME$(1)+" Error"+STR$(errCount)+" "+process$+" "+STR$(errorCode)
  GOSUB *sendLog
RETURN

*sendLog
  DO
    UDP_R1$=""
    UDP_X1$=sendStr$                                      /* Log PC に送信
    timer_=10                                             /* 0.1sec count down timer
    DO
      IF (UDP_R1$==sendStr$)|(timer_==0) THEN             /* エコーバックでハンドシェーク
        BREAK
      END_IF
    LOOP
    IF timer_<>0 THEN
      BREAK
    END_IF
    PR "sendLog Response Timeout"
  LOOP
RETURN

Q03UDECPU
IP 192.168.0.195
Port 10000

MPC-3000/3200

PC
IP 192.168.0.71
Port 9000

文字列をLogするUDPサーバーアプリ。
受信した文字列をループバックする。

LAN/WAN

・設定後の SET_IP 状態

SET_IP
IPadrs = 192.168.0.20
 MASK  = 255.255.255.0
Gateway= 192.168.0.248
OFF Delay ACK  baud=38400
MACadrs= 70B3D553D051
 WRITE_BULK Ok 7 10

1 : /  TCP   Telnet          23  Slave 
2 : +  UDP   PACKET       10000 <UDP_X0$><UDP_R0$>         ←PLDの通信チャンネル
3 : +  UDP   PACKET        9000 <UDP_X1$><UDP_R1$>         ←PCの通信チャンネル

・PC UDPサーバーのLog
午前10時～午後5時10万回以上でQ3Eエラーは12回。タイムアウト時間を延ばせばもっと少なくなるかも。

10:06:45 Begin
10:10:00 833回
10:20:00 3342回
10:30:00 5836回
10:40:00 8349回
10:40:33 Error1 WRITE_BULK 95
10:50:00 10864回
11:00:00 13371回
11:10:00 15890回
11:19:35 Error2 READ_BULK 95
11:20:00 18400回
11:30:00 20928回
11:40:00 23442回
11:50:00 25959回
12:00:00 28466回
12:10:00 30974回
12:20:00 33487回
12:25:27 Error3 READ_BULK 95
12:30:00 35992回
12:40:00 38505回
12:50:00 41021回
13:00:00 43536回
13:10:00 46044回
13:20:00 48562回
13:25:04 Error4 READ_BULK 95
13:30:00 51066回
13:40:00 53584回
13:50:00 56101回
13:52:01 Error5 WRITE_BULK 95
13:55:03 Error6 WRITE_BULK 95
13:57:41 Error7 WRITE_BULK 95
14:00:00 58602回
14:10:00 61121回
14:20:00 63636回
14:30:00 66143回
14:40:00 68653回
14:42:06 Error8 READ_BULK 95
14:50:00 71156回
14:59:15 Error9 WRITE_BULK 95
15:00:00 73669回
15:10:00 76193回
15:20:00 78712回
15:29:30 Error10 READ_BULK 95
15:30:00 81218回
15:40:00 83739回
15:50:00 86256回
16:00:00 88758回
16:10:00 91276回
16:15:47 Error11 READ_BULK 95
16:20:00 93786回
16:30:00 96304回
16:40:00 98817回
16:41:05 Error12 READ_BULK 95
16:50:00 101329回
17:00:00 103841回

